

# A2 Oberon Programming Quickstart Guide

Lecture Series on System Construction,

Felix Friedrich, Ulrike Glavitsch, Florian Negele, Sven Stauber

October 28, 2019

This is a brief introduction to the programming language Active Oberon and the runtime kernel A2. This introduction does neither replace a language report nor is it an introduction to programming. It rather should provide quick access to the programming language and its surroundings by ways of comments and examples. It has been initially prepared for the Operating System Course, Spring Semester 2009, at ETH Zurich as accompanying material to the practical exercises and kept up to date for several courses afterwards.

## Contents

<b>1</b>	<b>Why Another Development Environment?</b>	<b>3</b>
<b>2</b>	<b>The A2 Runtime System Kernel</b>	<b>3</b>
2.1	Modules and Commands . . . . .	3
2.2	Loading and Unloading Modules . . . . .	4
<b>3</b>	<b>The Language Active Oberon</b>	<b>5</b>
3.1	Module Structure . . . . .	5
3.1.1	Import Section . . . . .	5
3.1.2	Constant Declaration Section . . . . .	5
3.1.3	Type Declaration Section . . . . .	6
3.1.4	Variable Declaration Section . . . . .	7
3.1.5	Procedure Declaration Section . . . . .	7
3.1.6	Body . . . . .	9
3.2	Basic Types . . . . .	9
3.2.1	Numeric types . . . . .	10

*A2 Programming Quickstart Guide*

2

3.2.2	BOOLEAN . . . . .	11
3.2.3	SET . . . . .	11
3.2.4	CHAR . . . . .	11
3.3	Composite Types . . . . .	12
3.3.1	Arrays . . . . .	12
3.3.2	Records . . . . .	13
3.3.3	Objects . . . . .	13
3.3.4	Inheritance . . . . .	14
3.4	Concurrency Support . . . . .	15
3.4.1	Active Objects . . . . .	15
3.4.2	Protection . . . . .	15
3.4.3	Synchronization . . . . .	17
3.4.4	Examples . . . . .	18
<b>4</b>	<b>Application Programming Interfaces</b>	<b>18</b>
4.1	Streams . . . . .	19
4.2	Commands . . . . .	21
4.3	Files . . . . .	22
4.4	Strings . . . . .	25
<b>A</b>	<b>Appendix</b>	<b>26</b>
A.1	Built-in Functions . . . . .	26
A.2	The Module SYSTEM . . . . .	27
A.2.1	BIT Manipulation . . . . .	27
A.2.2	SYSTEM Types . . . . .	28
A.2.3	Unsafe Typecasts . . . . .	28
A.2.4	Direct Memory Access Functions . . . . .	28
A.3	Systems Programming with Oberon . . . . .	28
A.4	Active Oberon EBNF . . . . .	30

## 1 Why Another Development Environment?

You are probably not used to programming in both Oberon and its development environments. In the first moment you feel uncomfortable with the unusual way how programs<sup>1</sup> are written, started and executed. You miss fancy menus, code completion and code lookup features. Some certainly ask why we not use IDEs such as Eclipse or Visual Studio for the development.

There are obvious disadvantages when everything is built by a very limited number of people. Also, having to build and extend everything by yourself is a laborious task. However, it has very interesting advantages:

- By providing the full tool-chain yourself, you have full control. Extension and maintenance becomes simpler. Licensing issues do not appear.
- By having to provide everything by yourself, you learn and gain understanding over all levels. It becomes clear how design decisions at one level influence other levels.
- By having to use the tools you built yourself, you provide stability and make sure that software stays maintained. You constantly test your tools and make sure they are useful.

We hope that, having worked for a certain time with A2 and Oberon, you will start to understand and like the way things are done.

## 2 The A2 Runtime System Kernel

A2 is a modern, multi-processor operating system in the tradition of the ETH operating systems Medos, Oberon and AOS. It is written in Active Oberon with only very few exceptions: little parts of the kernel and some low-level optimized routines are implemented in assembler. A2 is composed of light-weight processes that are supported by *Active Objects* in the language Active Oberon. Active Oberon is a highly readable, abstract programming language with built-in concurrency support. Its abstraction protects programmers from artificial technical complexities while system-near programming is still possible by ways of a special module named 'SYSTEM'.

### 2.1 Modules and Commands

Active Oberon is a modular programming language that is supported by a dynamic loading concept in the operating system A2. The role of programs and applications is played by *modules*. Instead of loading and executing a 'program', *modules* are loaded and *commands* are executed. Once a module is loaded, it stays loaded until it is explicitly unloaded or until the system goes down. Before we give more details on modules and commands, we show an example of a very simple module.

---

<sup>1</sup>Strictly speaking, the notion of a program does even not exist in Oberon

```

MODULE Hello;

IMPORT Commands, KernelLog;

  (* a command taking a context parameter *)
  PROCEDURE World*(context: Commands.Context);
  BEGIN
    context.out.String("Hello World");
    context.out.Ln
  END World;

  (* a command without parameters *)
  PROCEDURE InnerWorld*;
  BEGIN
    KernelLog.String("Hello Kernel World");
    KernelLog.Ln;
  END InnerWorld;

END Hello.

```

Fig. 1: A Simple Oberon Module interfacing A2 Modules.

Commands are executed by activating a command string of the form `ModuleName.CommandName` within any text displayed on the screen. In the simple example above, this would be the strings `Hello.World` and `Hello.InnerWorld`. A command is usually activated by a middle-mouse click or by selecting the command string and using the key-combination `Ctrl+Return`. When a command is activated, the module with name `ModuleName` is dynamically loaded on demand and the procedure named `CommandName` in module `ModuleName` is executed.

Note that a module name has nothing to do with the name of the file that contains the source code of the module. It is pure convention that source code files are usually denominated as `ModuleName.Mod`.

## 2.2 Loading and Unloading Modules

When a module is compiled, an *object file* is generated and is made persistent in a file. It is not immediately loaded. Basically, a module `M` is loaded when a command of the form `M.CommandName` is activated. A module `X` is also automatically loaded when any module importing `X` is loaded (cf. Paragraph 3.1.1).

A module stays loaded until it is unloaded explicitly. Modules can be unloaded with the command

```
System.Free ModuleName1 ModuleName2 ... ~
```

where the tilde ‘~’ indicates the end of the list. Note that a module `A` can only be unloaded if no other currently loaded module imports `A`.

## 3 The Language Active Oberon

### 3.1 Module Structure

A module consists of an import list, constant declarations, type declarations, variable declarations, procedure declarations and a body, all optional. In this Section, we describe these parts.

#### 3.1.1 Import Section

The exported symbols of a module can be made available to other modules. To make the symbols of module A available in module B, module A has to be imported in module B. A module import graph is a directed graph where cycles are forbidden.<sup>2</sup>

Symbols in one module that are to be used in any other module must be exported. A symbol is exported by suffixing the name with an asterisk '\*'. It is exported read only by suffixing with a minus symbol '-'. We provide a little example of a module A being imported by a module B.

```
MODULE A;  
  CONST ExportedConst*=100;  
END A.  
  
MODULE B;  
  IMPORT A;  
  VAR c-: INTEGER;  
BEGIN  
  c := A.ExportedConst;  
END B.
```

Fig. 2: B imports A

#### 3.1.2 Constant Declaration Section

Although static values can be used in expressions, it is possible to associate an identifier with a constant expression, in other words: to *declare a constant*. Only values of basic type and strings<sup>3</sup> can be declared as constants<sup>4</sup>. The following module contains some examples.

```
MODULE Random;  
  
CONST
```

---

<sup>2</sup>Cycles in the import graph are detected by the compiler.

<sup>3</sup>In Active Oberon, strings are array of characters, cf. Section 3.3.1

<sup>4</sup>There are exceptions: special mathematical arrays can also be declared as constants but this goes beyond the scope of this short reference.

```

(* ----- global constants ----- *)
Max      = 2147483647;
Allbits  = 7FFFFFFFH; (* hex number *)

VAR
(* ----- global variable ----- *)
z: SIGNED32;

PROCEDURE Rand*(): SIGNED32;
(* constants can also be declared in procedure scope: *)
CONST A = 16807; Q = 127773; R = 2836;
VAR t: SIGNED32;
BEGIN
    t := A * (z MOD Q) - R * (z DIV Q);
    IF t > 0 THEN z := t ELSE z := t + Max END;
    RETURN z;
END Rand;
BEGIN
    z := Allbits;
END Random.

```

Fig. 3: Declaration and Usage of Constants

### 3.1.3 Type Declaration Section

Type declarations bind an identifier to a type that can be used in variable declarations to name the type of the variables (see section 3.1.4). Type declarations are used to provide synonyms for user-defined composite types or existing types such as all predeclared basic types (see sections 3.3 and 3.2 respectively). The following example uses a type declaration to assign a new name to a composite record type.

```

MODULE TypeDeclarations;

TYPE
    NumberType = REAL;
    ComplexNumber = RECORD re,im: NumberType END;

VAR
    res,a,b: ComplexNumber;

PROCEDURE Multiply(VAR res: ComplexNumber; x,y: ComplexNumber);
BEGIN
    res.re := x.re*y.re - x.im*y.im;
    res.im := x.re*y.im + x.im*y.re;
END Multiply;

BEGIN

```

```

    (* ... *)
    Multiply(res,a,b);
    (* ... *)
END TypeDeclarations.

```

Fig. 4: Using Type Declarations

Note that the call of procedure `Multiply` with variables `res`, `a` and `b` requires that the variables are assignment compatible to the respective parameters. This is guaranteed by using the name of a type declaration which by definition always refers to the same type.

### 3.1.4 Variable Declaration Section

Variables have to be declared before use in a variable declaration section. A variable declaration consists of an identifier and a type. Variables can be declared globally, in a procedure scope, in an object or in a record scope.

```

MODULE Variables;
TYPE
    (* variables in a record *)
    Pair = RECORD x,y: REAL END;
    PairPointer = POINTER TO Pair;
    ExampleObject = OBJECT
        (* object variables *)
        VAR x,y: REAL;
    END ExampleObject;

VAR (* global variables *)
    a: INTEGER;
    b: Pair;
    c: PairPointer;

    PROCEDURE Example;
    (* local variables *)
    VAR a,b: REAL;
    BEGIN
    END Example;

END Variables.

```

Fig. 5: Variables and Parameters

### 3.1.5 Procedure Declaration Section

Procedures contain the executable code of a module. They can be declared in a module scope, within procedure scopes ('nested procedures') or in object scopes ('methods'). Parameters are

special variables that refer to expressions that have been passed to the procedure from its caller. Parameters are declared in the head of the procedure declaration.

Parameters can be *value parameters*<sup>5</sup>, in which case they represent the actual value of an expression being passed to the respective procedure. Or parameters can be *variable parameters*<sup>6</sup>, in which case they stand for the reference of a designator that has been passed to the procedure. The modification of a value parameter within a procedure does only have a temporary effect within the procedure while the modification of a variable parameter implies the modification of the passed variable.

Parameters can also be *const parameters*. In contrast to variable or value parameters, const parameters cannot be modified within the procedure. Declaring a variable constant has the advantage that the compiler can optimize the way how parameters are passed. Composite types like records or arrays, for example, are passed as reference while basic types will usually be called by value.

```
MODULE Procedures;

TYPE
  O = OBJECT
    PROCEDURE Method;
    PROCEDURE NestedProcedure;
    BEGIN
      END NestedProcedure;
    BEGIN
      END Method;
  END O;

  PROCEDURE GlobalProcedure;
    PROCEDURE NestedProcedure;
    BEGIN END NestedProcedure;
  END GlobalProcedure;

  PROCEDURE Square(x: REAL): REAL;
  BEGIN x := x*x; RETURN x
  END Square;

  PROCEDURE Inc(VAR i: INTEGER);
  BEGIN i := i+1
  END Inc;

  PROCEDURE Print(CONST c: ComplexType);
  BEGIN
    ....
  END;
```

---

<sup>5</sup>Value parameters are passed over the stack.

<sup>6</sup>Variable parameters are references being passed over the stack.



```
PROCEDURE Test*;
VAR x, sq: REAL; i: INTEGER;
BEGIN
    x := 10; i := 0;
    sq := Square(x);
    Inc(i);
END Test;

END Procedures.
```

Fig. 6: Procedures

### 3.1.6 Body

A module can provide a body. The body will be executed once after the module has been loaded but prior to any of its commands. Consider the following example:

```
MODULE Hello;

IMPORT KernelLog;

PROCEDURE World*;
BEGIN KernelLog.String("Hello World"); KernelLog.Ln;
END World;

BEGIN
    (* this is a special 'procedure': the module body
       it is executed if and only if a module has been freshly loaded *)
    KernelLog.String("Hello has been loaded"); KernelLog.Ln;
END Hello.
```

Fig. 7: A simple Oberon Module

A first call of `Hello.World` results in the output `Hello has been loaded` and `Hello World` in the kernel log. Each subsequent call will result in an output of `Hello World`, but no further execution of the body. Only if the module has been unloaded, the body is executed again. Note that, consequently, in order to see the effects of a modification and re-compilation of a module it has to be unloaded. (cf. Section 2.2).

## 3.2 Basic Types

The predeclared basic types of Active Oberon are shown with names and valid values in Table 1.

Type name	Size	Valid values
BOOLEAN	1 byte	TRUE or FALSE
CHAR	1 byte	characters of the extended ASCII set (0X ... 0FFX)
SIGNED8	1 byte	integers between $-2^7$ and $2^7 - 1$
SIGNED16	2 bytes	integers between $-2^{15}$ and $2^{15} - 1$
SIGNED32	4 bytes	integers between $-2^{31}$ and $2^{31} - 1$
SIGNED64	8 bytes	integers between $-2^{63}$ and $2^{63} - 1$
UNSIGNED8	1 byte	integers between 0 and $2^8 - 1$
UNSIGNED16	2 bytes	integers between 0 and $2^{16} - 1$
UNSIGNED32	4 bytes	integers between 0 and $2^{32} - 1$
UNSIGNED64	8 bytes	integers between 0 and $2^{64} - 1$
FLOAT32	4 bytes	floating point value between $-3.4028^{38}$ and $+3.4028^{38}$
FLOAT64	8 bytes	floating point value between $-1.7976^{308}$ and $+1.7976^{308}$
SET8	1 byte	any set combination of the integer values between 0 and 7
SET16	2 bytes	any set combination of the integer values between 0 and 15
SET32	4 bytes	any set combination of the integer values between 0 and 31
SET64	8 bytes	any set combination of the integer values between 0 and 63
REAL		default floating point type corresponds to double in C
INTEGER	machine word	signed integers in machine word size, corresponds to <code>int</code> in C
ADDRESS	address width	unsigned integers in address range
SIZE	address width	signed integers in address range
SET	address width	set with address width

Table 1: Predeclared Basic Types

### 3.2.1 Numeric types

The *numeric types* are comprised of the integer and real types and form a set hierarchy

$$FLOAT64 \supset FLOAT32 \supset SIGNED64 \supset SIGNED32 \supset SIGNED16 \supset SIGNED8$$

The range of the larger type includes the ranges of the smaller types. The smaller type is said to be *compatible* with the larger one in the sense that it can without danger of loss of leading digits be converted. In assignments and in expansions the conversion of internal representations is automatic.

*Unsigned integers* are compatible with signed or unsigned integer of same or smaller size. This implies that the assignment from a signed to an unsigned integer of same size is ok. The other direction does not work:

Although the `SIZE` type is signed and `ADDRESS` is unsigned, `SIZE` and `ADDRESS` types are assignment compatible in both directions.

```
VAR
    s8: SIGNED8; s16: SIGNED16; s64: SIGNED64;
    u8: UNSIGNED8; u16: UNSIGNED16; u64: UNSIGNED64;
    adr: ADDRESS; size: SIZE;
BEGIN
    s16 := s8; (* ok *)
    u16 := s8; (* ok *)
    s16 := u8; (* ok *)
    u16 := s16; (* ok *)
    adr := size; (* ok *)
    size := adr; (* ok *)

    s16 := u16; (* error *)
```

Fig. 8: Signed vs. Unsigned Types

### 3.2.2 BOOLEAN

A Boolean value is one of the *logical truth values* which are represented in Active Oberon by the standard identifiers TRUE and FALSE.

### 3.2.3 SET

The values which belong to the type SET are *elements of the power set of*  $\{0, 1, \dots, N\}$  where  $N$  is equal to MAX(SET), a constant defined by the implementation. It is typically the address width of the computer. In fact, sets are efficiently implemented as bit operations.

Examples of set constants are:

```
MODULE SetConstants;

CONST
    EmptySet = {};
    SomeElements = {1, 6, 10};
    SomeOthers = {0, 2..4, 8};
END SetConstants.
```

Fig. 9: Using set constants

where {} denotes the empty set and the expression 2..4 refers to the elements 2, 3 and 4.

### 3.2.4 CHAR

A major portion of the input and output of computers is in the form of character strings that contain values of type CHAR. The value range of the type CHAR consists of the characters of the roman alphabet and a small set of special symbols used frequently in commerce and mathematics.

The set representing the type CHAR is ordered and each character has a fixed position or *ordinal number*. This is reflected in the notation for character constants which may be written as "a" or 61X for the letter a. The first representation denotes the value of the variable of type CHAR, the second its (hexadecimal) *ordinal* number.

### 3.3 Composite Types

In Active Oberon, there are basically three composite types available: Arrays, Records and Objects.

#### 3.3.1 Arrays

Arrays can be declared static, open or dynamic. Static arrays are declared as ARRAY number OF BaseType, where number must be constant. The BaseType can be an array, which allows the declaration of multi-dimensional arrays. Arrays can be declared open only in a parameter section. Dynamic arrays are basically references pointing to an array with lengths that may be provided during runtime.

```

MODULE Arrays;
VAR
  a: ARRAY 32 OF ARRAY 20 OF INTEGER; (* static, two-dimensional *)
  b: POINTER TO ARRAY OF INTEGER; (* dynamic, one-dimensional *)

PROCEDURE Print(x: ARRAY OF INTEGER); (* open *)
VAR i: INTEGER;
BEGIN
  i := 0;
  WHILE i < LEN(x) DO
    (* Printout x[i] *)
    INC(i);
  END;
END Print;

PROCEDURE Example;
BEGIN
  NEW(b, 100); (* allocate array *)
  (* ... *)
  Print(b^);
END Example;

END Arrays.

```

Fig. 10: Using arrays

### 3.3.2 Records

Records are containers of data. Records are value types and are declared as **RECORD** (variables) **END**. Records can also be declared as reference types using **POINTER TO**. Record fields are referred to via `recordName.variableName`.

```
MODULE Records;
VAR
  a: RECORD x,y: INTEGER END;
  p: POINTER TO RECORD a,b: REAL END;

BEGIN
  NEW(p);
  p.b := a.x;
END Records.
```

Fig. 11: Using records

### 3.3.3 Objects

Objects are basically *reference* records that can additionally be equipped with procedures. Procedures in an object are methods: they reside in the object scope and have access to the object's variables. An object can be explicitly referred to in its method using the **SELF** identifier. A method prefixed by an ampersand character **&** is an *object initializer*. This method is automatically called when an instance of the object is created and processed before the object becomes publicly available. An object may have at most one initializer. If absent, the initializer of the base type is inherited. Initializers can be called like methods.

```
MODULE Objects;
TYPE
  MyObject = OBJECT (* class *)
    VAR x,y: REAL;

    PROCEDURE Equals(o: MyObject): BOOLEAN;
    BEGIN
      IF o = SELF THEN RETURN TRUE
      ELSE RETURN (o.x = x) & (o.y = y)
      END;
    END Equals;

    PROCEDURE &Init(x, y : REAL); (* initializer *)
    BEGIN
      SELF.x := x; SELF.y := y;
    END Init;
  END MyObject;

VAR p,q: MyObject; (* objects / instances *)
```

```
BEGIN
  (* instantiate object p with x and y parameters for initializer *)
  NEW(p, 1.0, 9.99);
  (* ... *)
  IF p.Equals(q) THEN (* ... *)
  END;
END Objects.
```

Fig. 12: Usage of Objects

### 3.3.4 Inheritance

Active Oberon is an object oriented language. Inheritance is supported for records and objects. It is possible to extend records and objects and to use type checks, type guards and overriding of methods. Record and object extension is explicated as in the following example:

```
MODULE Extensions;

TYPE
Rectangle = RECORD
  x, y, w, h: INTEGER
END;

FilledRectangle = RECORD(Rectangle)
  color: INTEGER
END;

Window = OBJECT
VAR x, y, w, h: INTEGER;

  PROCEDURE Print;
  BEGIN
    (* draw frame *)
  END Print;
END Window;

FilledWindow = OBJECT (Window)
VAR color: INTEGER;

  PROCEDURE Print; (* overrides Window.Print *)
  BEGIN
    Print^; (* supercall *)
    (* fill *)
  END Print;
END FilledWindow;
```

**END** Extensions.

Fig. 13: Type extension of records and objects

### 3.4 Concurrency Support

Active Oberon provides built-in concurrency support. Threads are represented as *Active Objects* and language constructs for ensuring mutual exclusion and thread synchronization are provided.

#### 3.4.1 Active Objects

The declaration of an object type may include an *object body*. The body is the object's activity, to be executed whenever an object instance is allocated after the initializer (if any) completed execution. The object body is annotated with the *ACTIVE* modifier. At allocation, a new process is allocated to execute the body concurrently. If the *ACTIVE* modifier is not present, the body is executed synchronously, i.e. *NEW* returns only after the body has terminated execution. The active object activity terminates whenever the body execution terminates. As long as the body executes, the object is kept alive (in particular it cannot be garbage collected).

```
MODULE Example;
TYPE
  ActiveObject = OBJECT
    BEGIN {ACTIVE}
      (* do some useful work *)
    END ActiveObject;

VAR
  o : ActiveObject;

BEGIN
  (* instantiate active object o *)
  NEW(o);
END Example.
```

Fig. 14: Example of an active object

#### 3.4.2 Protection

Like a (procedure-, module- or object-) body, a Statement Block is a sequence of statements delimited by *BEGIN* and *END*. It can be used anywhere like a simple statement. An *EXCLUSIVE* modifier turns a statement block (or body) into a critical region to protect the statements against concurrent execution. Our protection model is an instance-based monitor: Every object instance is protected and the protection granularity is any statement block inside the object's method, ranging from a single statement to a whole method.

Upon entering an exclusive block, an activity is preempted as long as another activity stands in an exclusive block of the same object instance. An activity cannot take an object's lock more than once (re-entrance is not allowed). Modules are considered singleton objects, thus procedures in a module can also be protected. In this case the scope of protection is the whole module.

```
MODULE Demo;  
  
TYPE  
  SomeObject = OBJECT  
  
    PROCEDURE O1;  
    BEGIN {EXCLUSIVE}  
      (* critical section o1 *)  
    END O1;  
  
    PROCEDURE O2;  
    BEGIN  
      (* non-critical section *)  
      BEGIN {EXCLUSIVE} (* critical section o2 *) END;  
      (* non-critical section *)  
    END O2;  
  
  END SomeObject;  
  
  PROCEDURE P1;  
  BEGIN {EXCLUSIVE}  
    (* critical section p1 *)  
  END P1;  
  
  PROCEDURE P2;  
  BEGIN  
    (* non-critical section *)  
    BEGIN {EXCLUSIVE} (* critical section p2 *) END;  
    (* non-critical section *)  
  END P2;  
  
END Demo.
```

Fig. 15: Usage of exclusive blocks

In this example, at most one thread can be in one of the critical sections of the module *p1* or *p2* at any time. For each instance of *SomeObject*, at most one thread can be in *o1* or *o2* at any time. Note that there is no relation of *p1*, *p2* and *o1*, *o2* in different object instances since the protection is based on *instance-based* monitors.



### 3.4.3 Synchronization

The built-in procedure *AWAIT* is used to synchronize an activity with a state of the system. *AWAIT* can take any boolean condition. The activity is allowed to continue execution only when the condition is true. While the condition is not established, the activity remains suspended. The lock on the protected object is released, as long as the activity remains suspended<sup>7</sup>. The activity is resumed if and only if the lock can be taken. The system is responsible for evaluating the conditions and for restarting suspended activities. The conditions inside an object instance are re-evaluated whenever some activity leaves a protected block inside the same object instance. This implies that changing the state of an object outside a protected block does not imply a re-evaluation of its conditions. When several activities compete for the same object lock, the activities whose conditions are true are scheduled before those that only want to enter a protected region.

```
MODULE Example;

TYPE
  Synchronizer = OBJECT
    VAR awake : BOOLEAN;

    PROCEDURE &Init;
    BEGIN
      awake := FALSE;
    END Init;

    PROCEDURE Wait;
    BEGIN {EXCLUSIVE}
      AWAIT(awake); (* suspend caller until awake = TRUE *)
      awake := FALSE;
    END Wait;

    PROCEDURE Wakeup;
    BEGIN {EXCLUSIVE}
      awake := TRUE;
    END Wakeup;

  END Synchronizer;

END Example.
```

Fig. 16: Example of object synchronization

---

<sup>7</sup>Releasing the lock upon waiting allows other activities to change the state of the object and thus establish the condition

### 3.4.4 Examples

The following example shows the implementation of a bounded-buffer.

```
MODULE Example;  
  
TYPE  
  Item* = OBJECT END Item;  
  
  Buffer* = OBJECT  
  VAR head, num: SIZE; buffer: POINTER TO ARRAY OF Item;  
  
  PROCEDURE Append*(x: Item);  
  BEGIN {EXCLUSIVE}  
    AWAIT(num # LEN(buffer));  
    buffer[(head+num) MOD LEN(buffer)] := x;  
    INC(num)  
  END Append;  
  
  PROCEDURE Remove*(): Item;  
  VAR x: Item;  
  BEGIN {EXCLUSIVE}  
    AWAIT(num # 0);  
    x := buffer[head];  
    head := (head+1) MOD LEN(buffer);  
    DEC(num);  
    RETURN x  
  END Remove;  
  
  PROCEDURE &Init*(n: SIZE);  
  BEGIN  
    head := 0; num := 0; NEW(buffer, n)  
  END Init;  
END Buffer;  
  
END Example.
```

Fig. 17: Bounded-buffer

## 4 Application Programming Interfaces

This section comprises a brief overview of often used application programming interfaces.

## 4.1 Streams

Streams provide an abstraction for easily accessing various resources that can be accessed byte-wise. Streams can be opened on files, network connections, serial port connections, strings, memory and many other resources. Streams are unidirectional, i.e. it is possible to either read from a resource (*Readers*) or write to a resource (*Writers*). Readers maintain an input buffer and writers maintain an output buffer that has to be flushed explicitly by the programmer. Streams maintain an internal position that is updated automatically whenever a read or write operation is performed. Streams may support random access, depending on the underlying resource.

**MODULE** Streams;

Position\* = SIGNED64;

(\*\* A reader buffers input received from a Receiver.  
Must not be shared between processes. \*)

Reader\* = **OBJECT**

(\*\* result of last input operation \*)  
res\*: **INTEGER**;

(\*\* Current position. \*)  
**PROCEDURE** Pos\*( ): Position;

(\*\* Returns TRUE if this stream supports random access \*)  
**PROCEDURE** CanSetPos\*( ): **BOOLEAN**;

(\*\* Set position to <pos> (only if resource supports random access) \*)  
**PROCEDURE** SetPos\*( pos: **SIZE** );

(\*\* Return number of bytes currently available in input buffer. \*)  
**PROCEDURE** Available\*( ): **SIZE**;

(\*\* Read one byte. \*)  
**PROCEDURE** Char\*( **VAR** x: **CHAR** );

(\*\* Read one byte but leave the byte in the input buffer. \*)  
**PROCEDURE** Peek\*( ): **CHAR**;

(\*\* Read size bytes into x, starting at ofs.  
The len parameter returns the number of bytes that were actually  
read. \*)  
**PROCEDURE** Bytes\*( **VAR** x: **ARRAY OF CHAR**; ofs, size: **SIZE**; **VAR** len: **SIZE**  
);

(\*\* Skip n bytes on the reader. \*)  
**PROCEDURE** SkipBytes\*( n: **SIZE** );

```

(** -- Read formatted data (uses Peek for one character lookahead) --
*)

(** Read an integer value in decimal or hexadecimal.
    If hex = TRUE, recognize the "H" postfix for hexadecimal numbers. *)
PROCEDURE Int*( VAR x: SIGNED32; hex: BOOLEAN );

(** Read all characters until the end of the line (inclusive).
    If the input string is larger than x, read the full string and
    assign the truncated 0X-terminated value to x. *)
PROCEDURE Ln*( VAR x: ARRAY OF CHAR );

(** Skip over all characters until the end of the line (inclusive). *)
PROCEDURE SkipLn*;

(** Skip over space, TAB and EOLN characters. *)
PROCEDURE SkipWhitespace*;

(** Read an optionally '"' or "'" enquoted string.
    Will not read past the end of a line. *)
PROCEDURE String*( VAR string: ARRAY OF CHAR );

(** First skip whitespace, then read string *)
PROCEDURE GetString*(VAR string : ARRAY OF CHAR);

(** First skip whitespace, then read integer *)
PROCEDURE GetInteger*(VAR integer : SIGNED32; hex : BOOLEAN);

END Reader;

Writer* = OBJECT

(** result of last output operation *)
res*: INTEGER;

(** Current position. *)
PROCEDURE Pos*( ): SIZE;

(** Returns TRUE if this stream supports random access *)
PROCEDURE CanSetPos*( ): BOOLEAN;

(** Set position to <pos> (only if resource supports random access) *)
PROCEDURE SetPos*( pos: SIZE );

(** Flush output buffer *)
PROCEDURE Update*;

(** Write one byte. *)

```

```

PROCEDURE Char*( x: CHAR );

(** Write len bytes from x, starting at ofs. *)
PROCEDURE Bytes*(CONST x: ARRAY OF CHAR; ofs, len: SIZE );

(** -- Write formatted data -- *)

(** Write an ASCII end-of-line (CR/LF). *)
PROCEDURE Ln*;

(** Write a 0X-terminated string, excluding the 0X terminator. *)
PROCEDURE String*(CONST x: ARRAY OF CHAR );

(** Write an integer in decimal
    right-justified in a field of at least w characters. *)
PROCEDURE Int*( x: SIGNED64; w: SIZE );

(** Write an integer in hexadecimal
    right-justified in a field of at least ABS(w) characters. *)
PROCEDURE Hex*(x: UNSIGNED64; w: SIZE );

END Writer;

END Streams;

```

Fig. 18: Simplified Reader and Writer Interface

## 4.2 Commands

Commands are procedures that can be invoked directly by the command interpreter. As has already been indicated in Section 2.1, a procedure can act as a command if and only if it is exported, if it is declared in the module scope and if it has a command signature:

```

(** Command with no arguments *)
PROCEDURE ProcedureName*;

(** Command with arguments and output stream *)
PROCEDURE ProcedureName*(context : Commands.Context);

```

Fig. 19: Command Signatures

An excerpt of the Commands module interface reads as follows:

```

MODULE Commands;
TYPE
  Context* = OBJECT
  VAR
    in-, arg- : Streams.Reader;
    out-, error- : Streams.Writer;
    caller- : OBJECT;

```

```

END Context;
END Command;

```

Fig. 20: Command Context

Command line arguments can be accessed using the *arg* stream. The stream *in* is the command input stream. For output, the streams *out* and *error* are used. The caller field optionally contains a reference to an object that is responsible for the command invocation. Note that both output stream buffers (*out* and *error*) are flushed automatically by the command interpreter.

```

MODULE Example;
IMPORT Commands;

PROCEDURE Add*(context : Commands.Context);
VAR a, b : SIGNED32;
BEGIN
  context.arg.GetInteger(a, FALSE);
  context.arg.GetInteger(b, FALSE);
  context.out.Int(a, 0); context.out.String(" + "); context.out.Int(b,
    0);
  context.out.String(" = "); context.out.Int(a + b, 0);
  context.out.Ln;
END Add;
END Example.

```

Fig. 21: Command Example

After compilation of module Example, the command interpreter is able to process the command `Example.Add 3 5 ~`. Note that the tilde character indicates the end of the command line arguments.

### 4.3 Files

Module Files provides both the interface to be implemented by file system drivers and A2 file API.

The most useful operations are

```

MODULE Files;
TYPE
  File* = OBJECT;

  (** Open file <filename>. Returns NIL if file cannot be opened *)
  Old*(filename : ARRAY OF CHAR) : File;

  (** Create file <filename>. Returns NIL if file cannot be created *)
  New*(filename : ARRAY OF CHAR) : File;

```

```

(** Register a new file (create entry in file system directory) *)
Register*(f : File);
END Files;

```

Fig. 22: Basic Files API

The actual access to files is achieved by using the low-level file riders or by using streams on a file. We describe both methods in the subsequent sections.

**File Interface** The low-level file interface uses so-called riders as context for accessing files. Multiple riders can be positioned independently on a given file. Essentially, riders are used for tracking the position in a file and store the result of the last operation on the file. As for streams, read and write operations automatically update the current position.

```

Positon** = Streams.Position;
Size* = Streams.Position;

(** A rider points to some location in a file,
    where reading and writing will be done. *)
Rider* = RECORD (** not shareable between multiple processes *)
(** has end of file been passed *)
eof*: BOOLEAN;

(** leftover byte count for ReadBytes/WriteBytes *)
res*: INTEGER;
END;

File* = OBJECT (** sharable *)
(* ... *)

(** Position a Rider at a certain position in a file.
    Multiple Riders can be positioned at different locations in a file.
    A Rider cannot be positioned beyond the end of a file. *)
PROCEDURE Set*(VAR r: Rider; pos: Position);

(** Return the offset of a Rider positioned on a file. *)
PROCEDURE Pos*(VAR r: Rider): Position;

(** Read a byte from a file, advancing the Rider one byte further.
    R.eof indicates if the end of the file has been passed. *)
PROCEDURE Read*(VAR r: Rider; VAR x: CHAR);

(** Read a sequence of len bytes into the buffer x at offset ofs,
    advancing the Rider. Less bytes will be read when reading over the
    end of the file. r.res indicates the number of unread bytes. *)

PROCEDURE ReadBytes*(VAR r: Rider; VAR x: ARRAY OF CHAR; ofs, len:
    SIZE);

```

```

(** Write a byte into the file at the Rider position,
    advancing the Rider by one. *)
PROCEDURE Write*(VAR r: Rider; x: CHAR);

(** Write the buffer x containing len bytes (starting at offset ofs)
    into a file at the Rider position. *)
PROCEDURE WriteBytes*(VAR r: Rider; CONST x: ARRAY OF CHAR; ofs, len:
    SIZE);

(** Return the current length of a file. *)
PROCEDURE Length*(): Size;

(** Flush the changes made to a file from its buffers. *)
PROCEDURE Update*;

END File;

```

Fig. 23: Low-level Files API

**Using Streams On Files** First we display a part of the Files interface:

```

MODULE Files;
TYPE
  Positon** = Streams.Position;
  Size* = Streams.Position;
  File* = OBJECT; (* as described in last section *)
  Reader* = OBJECT (Streams.Reader);
  Writer* = OBJECT (Streams.Writer);

  OpenReader*(VAR r: Reader; f : File; position : Position);
  OpenWriter*(VAR w : Writer; f : File; position : Position);
END Files.

```

Fig. 24: Using Streams On Files

Next we provide an example of how to use streams on files.

```

MODULE Example;
IMPORT Commands, Files;

PROCEDURE CreateFile*(context : Commands.Context);
VAR
  filename : Files.FileName;
  file : Files.File; writer : Files.Writer;
  ch : CHAR;
BEGIN
  context.arg.GetString(filename);
  file := Files.New(filename);
  IF (file # NIL) THEN

```



```

Files.OpenWriter(writer, file, 0);
context.arg.Char(ch); (* skip argument delimiter character *)
REPEAT
    context.arg.Char(ch);
    writer.Char(ch);
UNTIL (ch = 0X);
writer.Update;
Files.Register(file);
END;
END CreateFile;

END Example.

```

Fig. 25: Example: Create A New File

In this example, executing the command `Example.CreateFile HelloWorld.txt Hello World ~` would create (or overwrite) the file "HelloWorld.txt" and write the 0X-terminated string "Hello World " into it.

## 4.4 Strings

The module Strings provides procedures for ASCII string manipulation.

```

MODULE Strings;

(** returns the length of a string *)
PROCEDURE Length* (CONST string: ARRAY OF CHAR): SIZE;

(** appends appendix to s: s := s || appendix *)
PROCEDURE Append* (VAR s: ARRAY OF CHAR; CONST appendix: ARRAY OF CHAR);

(** concatenates s1 and s2: s := s1 || s2 *)
PROCEDURE Concat* (CONST s1, s2: ARRAY OF CHAR; VAR s: ARRAY OF CHAR);

(** converts an integer value to a string *)
PROCEDURE IntToStr*(i: SIGNED64; VAR s: ARRAY OF CHAR);

(** Simple pattern matching with support for "*" and "?" wildcards
    returns TRUE if name matches mask. *)
PROCEDURE Match*(CONST mask, name: ARRAY OF CHAR): BOOLEAN;

END Strings.

```

Fig. 26: Strings

## A Appendix

### A.1 Built-in Functions

There are some built-in procedures and functions in Active Oberon. We give a short overview in the following table. Note that Integer stands for any integer type

Function	Argument Types	Result Type	Description
INC(x)	x: Integer		increment x by 1
DEC(x)	x: Integer		decrement x by 1
INC(x,n)	x: Integer; n: Integer		increment x by n
DEC(x,n)	x: Integer; n: Integer		decrement x by n
ASSERT(x)	x: BOOLEAN		assert trap, if x not true
COPY(x,y)	x,y: ARRAY OF CHAR		0X-terminated copy of x to y
INCL(s,e)	s: SET, e: Integer		include element e in set s
EXCL(s,e)	s: SET, e: Integer		exclude element e from set s
HALT(n)	n: Integer		generate a trap with number n
NEW(x,...)	x: Object or Pointer		allocate x
ABS(x)	x: Number	Number	return absolute value of x
SHL(x,y)	x,y: Integer	Integer	return shift left of x by y bits.
SHR(x,y)	x,y: Integer	Integer	return shift right of x by y bits. Type of x determines if this is logical or arithmetic shift
ROL(x,y)	x,y: Integer	Integer	return rotate left of x by y bits.
ROR(x,y)	x,y: Integer	Integer	return rotate right of x by y bits.
ASH(x,y)	x,y: Integer	Integer	return arithmetic shift of x by y bits (shifts right for n < 0)
LSH(x,n)	x: Integer; n: SIGNED32	Integer	Returns value x logically shifted left n bits (shifts right for n < 0)
ROT(x,n)	x: Integer; n: SIGNED32	Integer	Returns value x rotated left by n bits (rotates right for n < 0)
CAP(x)	x: CHAR	CHAR	return capital letter of x
CHR(x)	x: Integer	CHAR	return character with ascii-number x
ENTIER(x)	x: FLOAT32 or FLOAT64	SIGNED32	return largest integer not greater than x
LEN(x)	x: ARRAY OF CHAR	SIZE	return length of x

Function	Argument Types	Result Type	Description
MAX(t)	t: Number or SET	Number	return maximal number of basic type t
MIN(t)	t: Number or SET	Number	return minimal number of basic type t
ODD(x)	x: Integer	BOOLEAN	return if x is odd
ORD(x)	x: CHAR	SIGNED16	return ascii-number of x
SHORT(x)	x: Number	Number	number conversion down
LONG(x)	x: Number	Number	number conversion up
ADDRESSOF(v)	v: ANY	ADDRESS	returns the address of v
ADDRESS OF v	v: ANY	ADDRESS	returns the address of v
SIZEOF(T)	v: ANY	SIZE	returns the size of type T

The number conversion routines SHORT and LONG operate with respect to the relations

$$FLOAT64 \supset FLOAT32 \text{ and } SIGNED64 \supset SIGNED32 \supset SIGNED16 \supset SIGNED8.$$

## A.2 The Module SYSTEM

The (pseudo-)module SYSTEM contains definitions that are necessary to directly refer to resources particular to a given computer and/or implementation. These include facilities for accessing devices that are controlled by the computer, and facilities to override the data type compatibility rules otherwise imposed by the language definition. The functions and procedures exported by this module should be used with care! It is recommended to restrict their use to specific low-level modules. Such modules are inherently non-portable and easily recognized due to the identifier SYSTEM appearing in their import list. The subsequent definitions are applicable to the A2 operating system.

### A.2.1 BIT Manipulation

Function	Argument Types	Result Type	Description
BIT(adrr,n)	adr: ADDRESS; n: INTEGER	BOOLEAN	Returns TRUE if bit n at adr is set, FALSE otherwise

**A.2.2 SYSTEM Types**

Type	Description
BYTE	Representation of a single byte.

**A.2.3 Unsafe Typecasts**

Function	Argument Types	Result Type	Description
VAL(T,x)	T: Type; x: ANY	T	Unsafe type cast. Returns x interpreted as type T with no conversion

**A.2.4 Direct Memory Access Functions**

Function	Argument Types	Result Type	Description
PUT(adr,x)	adr: ADDRESS; x: Type		Mem[adr] := x where the size of type x is 8, 16, 32 or 64 bits
PUT8(adr,x)	adr: ADDRESS; x: (UN)SIGNED8		Mem[adr] := x
PUT16(adr,x)	adr: ADDRESS; x: (UN)SIGNED16		
PUT32(adr,x)	adr: ADDRESS; x: (UN)SIGNED32		
PUT64(adr,x)	adr: ADDRESS; x: (UN)SIGNED64		
GET(adr,x)	adr: ADDRESS; VAR x: Type		x := Mem[adr] where the size of type x is 8, 16, 32 or 64 bits
GET8(adr)	adr: ADDRESS	SIGNED8	RETURN Mem[adr]
GET16(adr)	adr: ADDRESS	SIGNED16	
GET32(adr)	adr: ADDRESS	SIGNED32	
GET64(adr)	adr: ADDRESS	SIGNED64	
MOVE(src, dst,n)	dst: ADDRESS; n: SIZE		Copy "n" bytes from address "src" to address "dst"

**A.3 Systems Programming with Oberon**

The most often observed problem of systems programmers new to Oberon is the lack of logical operators on integer numbers. There is no `a & b` and no `a | b`. Furthermore, bit-shifts have to be expressed either using functions such as `LSH` or `ASH` or using division. The following strategies are proposed for newbies in system programming with Oberon.

- Use `DIV` and `Multiplication` when you want to shift integer numbers.
- Use `LSH` when it is important that the sign bit is not propagated for right-shifts.
- Use `MOD` for masking lower bits.
- Use set operations when you want to operate on bit masks.
- Use `ODD` for bit-tests on integers.
- Use `SYSTEM.MOVE` when you want to copy large areas of memory.
- Use `SYSTEM.PUT` or `SYSTEM.GET` to read or write data byte-wise.
- Use `POINTER {UNSAFE} TO ...` when you want to access memory directly. Values compatible to type `ADDRESS` can be assigned to unsafe pointers.

Procedures, types and symbols can be flagged with special properties that influence the behavior of compiler and linker:

- Use `PROCEDURE {PLAIN} X(...)` to avoid that procedure `X` contains a procedure activation frame. Procedures marked `NOPAF` cannot provide variables or parameters.
- Use `PROCEDURE {OPENING} X(...)` to declare that a procedure should be linked first in an image. An opening procedure is always `PLAIN`.
- Use `PROCEDURE {CLOSING} X(...)` to declare that a procedure should be linked after all module bodies in an image. A closing procedure is always `PLAIN`.
- Use `VAR a {UNTRACED}: POINTER ...` in order to declare a pointer that is not traced by the Garbage Collector.
- Use `a {ALIGNED(32)}: ...` in order to make sure a symbol `a` gets aligned in memory accordingly.
- Use `x {OFFSET(100000H)}: ...` in order to make sure a symbol `x` gets pinned in memory accordingly.
- Use `BEGIN {UNCHECKED} ... END` in order to emit code without any checks such as stack-, null-pointer- or index bound checks.
- Use `POINTER {UNSAFE} TO RECORD` or `POINTER {UNSAFE} TO ARRAY` to declare a pointer that is inherently unsafe. An unsafe pointer is assignment compatible with an address. Clearly, unsafe pointers cannot be type guarded or checked. Unsafe pointer to open arrays have no length and cannot be passed as normal array.

Finally, you can write entire procedures or portions thereof using assembler.

- Use `CODE ... END` in order to write inline assembler code

Example:

```
PROCEDURE {OPENING} KernelBegin;
CODE
    MOV bootFlag, EAX
    LEA EAX, initRegs
    MOV [EAX + 0], ESI
    MOV [EAX + 4], EDI
END KernelBegin;
```

Fig. 27: Code that is linked to the front of a kernel image.

## A.4 Active Oberon EBNF

We display the syntax of Active Oberon in the Extended Backus-Naur-Form (EBNF). We present productions (syntactic equations) as equations with a single equal sign =. On the left hand side of a production stands the defined nonterminal symbol, the right hand side contains the substitution rule and is terminated by a period. Terminal symbols are embraced by single or double quotes (for example ':=', '"' and 'begin'). An alternative in a production is denoted by a vertical bar |. Brackets [ and ] denote optionality of the enclosed expression, while braces { and } denote its repetition (possibly 0 times). Additionally, parentheses ( and ) are used to enclose expressions and thereby control additional precedence.

```
Module = 'MODULE' [TemplateParameters] Identifier ['IN' Identifier] ';'
        {ImportList} DeclarationSequence [Body]
        'END' Identifier '.'.

TemplateParameters = '(' TemplateParameter {' TemplateParameter} ')'.

TemplateParameter = ('CONST' | 'TYPE') Identifier.

ImportList = 'IMPORT' Import { ',' Import } ';'.

Import = Identifier [':=' Identifier] ['(' ExpressionList ')'] ['IN' Identifier].

DeclarationSequence = {
    'CONST' [ConstDeclaration] {';' [ConstDeclaration]}
  | 'TYPE' [TypeDeclaration] {';' [TypeDeclaration]}
  | 'VAR' [VariableDeclaration] {';'
    [VariableDeclaration]}
  | ProcedureDeclaration
  | OperatorDeclaration
  | ';'
}

ConstantDeclaration = [IdentifierDefinition '=' ConstantExpression].

ConstantExpression = Expression.

VariableDeclaration = VariableNameList ':' Type.

VariableNameList = VariableName { ',' VariableName}.

VariableName = IdentifierDefinition [Flags]
               [':=' Expression | 'EXTERN' String].
```

```

Flags = '{' [ Flag {',' Flag} ] '}'.

Flag = Identifier ['(' Expression ')' | '=' Expression].

ProcedureDeclaration = 'PROCEDURE' ['^'|'&'|'~'|'-'|Flags ['-']]
    ['(' ParameterDeclaration ')']
    IdentifierDefinition [FormalParameters] ';'
    DeclarationSequence [Body]
    'END' Identifier.

OperatorDeclaration = 'OPERATOR' [Flags] ['-'] String ['*'|'-'] FormalParameters ';'
    DeclarationSequence
    [Body]
    'END' String.

FormalParameters = '(' [ParameterDeclaration {';' ParameterDeclaration}] ')'
    [':' [Flags] Type].

ParameterDeclaration = ['VAR'|'CONST'] Identifier [Flags] ['=' Expression]
    {'',' Identifier [Flags] ['=' Expression]}
    ':' Type.

Body = 'BEGIN' [Flags] StatementSequence ['FINALLY' StatementSequence]
    | 'CODE' Code.

TypeDeclaration = IdentifierDefinition '=' Type ';'.

Type = ArrayType | MathArrayType | RecordType | PointerType | ObjectType
    | ProcedureType | EnumerationType | QualifiedIdentifier
    | CellType | CellnetType | PortType.

ArrayType = 'ARRAY' [Expression {',' Expression}] 'OF' Type.

MathArrayType = 'ARRAY' '[' MathArraySize {',' MathArraySize} ']' 'OF' Type.

MathArraySize = Expression | '*' | '?'.

RecordType = 'RECORD' ['(' QualifiedIdentifier ')']
    [VariableDeclaration {';' VariableDeclaration}]
    {ProcedureDeclaration [';']| OperatorDeclaration [';']}
    'END'.

PointerType = 'POINTER' [Flags] 'TO' Type.

ProcedureType = 'PROCEDURE' [Flags] [FormalParameters].

ObjectType = 'OBJECT'
    | 'OBJECT' [Flags] ['(' QualifiedIdentifier ')']
    DeclarationSequence
    [Body]
    'END' [Identifier].

EnumerationType = 'ENUM' ['(' QualifiedIdentifier ')']
    IdentifierDefinition ['=' Expression]
    {'',' IdentifierDefinition ['=' Expression]}
    'END'.

CellType = ('CELL' | 'CELLNET') [Flags] ['(' PortList ')'] [';'] {ImportList}
    DeclarationSequence
    [Body] 'END' [Identifier].

```

## A2 Programming Quickstart Guide

32

```

PortList = [PortDeclaration {';' PortDeclaration}].

PortDeclaration = Identifier [Flags] {',' Identifier [Flags]} ':' PortType.

PortType = 'PORT' ('IN' | 'OUT') ['(' Expression ')']

QualifiedIdentifier = Identifier ['.' Identifier].

IdentifierDefinition = Identifier ['*' | '-' ].

Statement = [
  UnaryExpression
  [':= Expression
  | '!' Expression | '?' Expression | '<<' Expression | '>>' Expression
  ]
  | 'IF' Expression 'THEN' StatementSequence
  { 'ELSIF' Expression 'THEN' StatementSequence }
  [ 'ELSE' StatementSequence ]
  'END'
  | 'WITH' Identifier ':' QualifiedIdentifier 'DO' StatementSequence
  { '|' QualifiedIdentifier 'DO' StatementSequence }
  [ 'ELSE' StatementSequence ]
  'END'
  | 'CASE' Expression 'OF' ['|'] Case
  { '|' Case }
  [ 'ELSE' StatementSequence ]
  'END'
  | 'WHILE' Expression 'DO'
  StatementSequence
  'END'
  | 'REPEAT'
  StatementSequence
  'UNTIL' Expression
  | 'FOR' Identifier ':= Expression 'TO' Expression ['BY' Expression] 'DO'
  StatementSequence
  'END'
  | 'LOOP' StatementSequence 'END'
  | 'EXIT'
  | 'RETURN' [Expression]
  | 'AWAIT' Expression
  | StatementBlock
  | 'CODE' {any} 'END'
  | 'IGNORE' Expression
].

Case = RangeExpression {',' RangeExpression} ':' StatementSequence.

StatementBlock = 'BEGIN' [Flags] StatementSequence 'END'.

StatementSequence = Statement {';' Statement}.

Expression = RangeExpression [RelationOp RangeExpression].

RelationOp = '=' | '#' | '<' | '<=' | '>' | '>=' | 'IN' | 'IS'
            | '.=' | '.#' | '.<' | '.<=' | '.>' | '.>='
            | '??' | '!!' | '<<?' | '>>?'.

RangeExpression = SimpleExpression
  | [SimpleExpression] '..' [SimpleExpression] ['by' SimpleExpression]
  | '*' .

```



## A2 Programming Quickstart Guide

33

```

SimpleExpression = Term {AddOp Term}.

AddOp             = '+' | '-' | 'or'.

Term = Factor {MulOp Factor}.

MulOp = '*' | '/' | 'DIV' | 'MOD' | '&'
       | '.*' | './' | '\' | '**' | '+*' .

Factor = UnaryExpression | UnaryOperation Factor.

UnaryOperation = '~' | '+' | '-'.

UnaryExpression = PrimaryExpression [DesignatorOperations] [Flags].

PrimaryExpression = Number | Character | String | Set | Array
                  | 'NIL' | 'IMAG' | 'TRUE' | 'FALSE' |
                  | 'SELF' | 'RESULT' | 'ADDRESS' | 'SIZE'
                  | 'SIZE' 'OF' Factor | 'ADDRESS' 'OF' Factor
                  | 'ALIAS' OF Factor
                  | 'NEW' QualifiedIdentifier '(' ExpressionList ')'
                  | '(' Expression ')'
                  | Identifier.

DesignatorOperations = { "(" [ExpressionList] ")"
                       | "." Identifier
                       | '[' IndexList ']'
                       | '^'
                       | '\'
                       } .

ExpressionList = Expression { ',' Expression }.

IndexList = '?' [',' ExpressionList]
          | ExpressionList [',' '?' [',' ExpressionList] ]

Array = '[' Expression {',' Expression} ']'.

Set = "{" [RangeExpression {"," RangeExpression}] "}".

String = '"' {Character} '"' | "'" {Character} "'" | '\'" {Character} '\'.

Number = Integer | Real.

Integer = Digit {[""]Digit} | Digit {[""]HexDigit} 'H'
        | '0x' {[""]HexDigit} | '0b' {[""]BinaryDigit}.

Real = Digit {[""]Digit} '.' {Digit} [ScaleFactor].

ScaleFactor = ('E' | 'D') ['+' | '-'] digit {digit}.

HexDigit = Digit | 'A' | 'B' | 'C' | 'D' | 'E' | 'F'
          | 'a' | 'b' | 'c' | 'd' | 'e' | 'f' .

Identifier = Letter {Letter | Digit | '_' }.

Letter = 'A' | 'B' | .. | 'Z' | 'a' | 'b' | .. | 'z' .

Digit = '0' | '1' | '2' | '3' | '4' | '5' | '6' | '7' | '8' | '9'.

```